

01768C PROGRAM DESCRIPTION I

Page 1 of 2

Program Title SUPERELEVATED ROADWAY DESIGN WITH ELEVATIONS

Designed and developed by US A
Software Project of US A

Contributor's Name Greg Laragan

On Cards

Address 3530 11th Street #17

City Lewiston

State/Country Idaho

Zip Code 83501

Program Description, Equations, Variables The SUPER program solves for the centerline and shoulder elevations of a superelevated roadway curve of a normal crown roadway section. The solution is for superelevation attained by rotation about the roadway centerline, and the transition from crown section to superelevated section is made without the use of spirals.

The input phase of the program is divided into two sections: "Vert Align" and "Horiz Align". The "Vert Align" section is patterned after the HP Surveying Module "VERT" program, and input prompts and required data are exactly the same as for the "VERT" program. The SUPER program has been developed to allow use of the "VERT" program "*" subroutines if the Surveying Module is in place. This enables the program to run with one less memory module than is required without the Surveying Module. The "*" subroutines are included as a separate program in case the Surveying

Necessary Accessories 3 memory modules*. Printer optional; card reader helpful.

Operating Limits and Warnings The user must be aware of changes in alignment that could cause bad output data, i.e. a grade change or new horizontal or vertical curve. The program assumes that if a printer is connected it will be turned on.

*Only two memory modules are required if the HP Surveying Module is used.

Reference(s) American Association of State Highway Officials. A Policy on Geometric Design of Rural Highways. Washington D.C. 1966

Idaho Transportation Department, Division of Highways. Survey and Plans Manual.

This program has been verified only with respect to the numerical example given in *Program Description II*. User accepts and uses this program material AT HIS OWN RISK, in reliance solely upon his own inspection of the program material and without reliance upon any representation or description concerning the program material.

NEITHER HP NOR THE CONTRIBUTOR MAKES ANY EXPRESS OR IMPLIED WARRANTY OF ANY KIND WITH REGARD TO THIS PROGRAM MATERIAL, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE. NEITHER HP NOR THE CONTRIBUTOR SHALL BE LIABLE FOR INCIDENTAL OR CONSEQUENTIAL DAMAGES IN CONNECTION WITH OR ARISING OUT OF THE FURNISHING, USE OR PERFORMANCE OF THIS PROGRAM MATERIAL.

Module is not available.

The "Vert Align" section prompts until sufficient parameters have been entered to define either a straight grade vertical alignment or a vertical curve connecting two different grades. Once the alignment has been defined, the centerline elevation of any station within the range of the defined alignment can be calculated. The SUPER program contains modifications to the "VERT" program to also allow calculation of stations outside the limits of vertical curves, provided the alignment either side of the curve remains as initially defined. If a grade change or new vertical curve is encountered, the "Vert Align" input must be changed.

The "Horiz Align" section of the input phase is required to establish the limits of horizontal curvature and to determine the necessary superelevation and tangent runout lengths for the given design speed, superelevation rate and shoulder widths. Because the horizontal alignment is generally firmly established at the time superelevation grade data is required, the program assumes that all of the necessary horizontal alignment parameters have been previously calculated and therefore the "Horiz Align" inputting does not have the flexibility of the "Vert Align" section.

The program allows different shoulder widths either side of centerline and makes the calculation of runoff length(RL) to accomodate the shoulder that is the greatest distance from the axis of rotation, which in this case is the "centerline". This way the runoff length provides the smoothest transition from normal crown section to full superelevated section, regardless of differences in lane widths. The runoff length is calculated using the input values for design speed, superelevation rate and shoulder widths, and the calculated value is compared against the minimum acceptable value for the given design speed. If the calculated runoff length is less than the minimum standard, the minimum value is used; otherwise the calculated value is used and rounded to the nearest multiple of 10 or 25. The runoff length, superelevation rate and crown rate are then used to calculate the corresponding tangent runout length(Z), and RL and Z are displayed. The runoff and runout length can be calculated independently of the other program sections without the need for input of any vertical alignment data by executing the subroutine "HA".

Prior to the initial calculation of any roadway grades, it is a good idea to check the superelevation rate to verify that it is adequate for the design speed and the degree of curve. The SUPER program provides this check through a subroutine that calculates the safe speed and compares it to the design speed, adjusts the design speed to equal the calculated safe speed, calculates a new safe speed based on the adjusted design speed, and continues the process until the two values are within 0.5 miles per hour of each other.

Elevation data can be obtained for any individual location within the defined alignment through the use of the G key as in the "VERT" program, or after the initial station has been entered, a specific interval for spacing of points can be input by using the i key, again as in the "VERT" program. The SUPER program will then prompt for the last station desired for output. This allows calculation of grades beyond the end of vertical curves.

The program calculates the shoulder elevations by first calculating and storing the centerline elevation at the input station, and then determining the location of the station with respect to the superelevation rotation. Any station must satisfy one of the following cases: (1) Located at either end of the horizontal curve, before or after any tangent runout occurs. This would be: Station \leq (PC-2RL/3-Z) or Station \geq (PT+2RL/3+Z); (2) Located within the tangent runout at either end of the curve: (PC-2RL/3-Z) \leq Station \leq (PC-2RL/3+Z) or (PT+2RL/3-Z) \leq Station \leq (PT+2RL/3+Z); (3) Located at either end of the curve in a partially superelevated section: (PC-2RL/3+Z) \leq Station \leq (PC+RL/3) or (PT-RL/3) \leq Station \leq (PT+2RL/3-Z); (4) Located within the fully superelevated section: (PC+RL/3) \leq Station \leq (PT-RL/3). Once the program determines which range the station is within, it calculates the proper values to be added to or subtracted from the centerline elevation to get the shoulder elevations. The decision to add or subtract is controlled by the status of Flag 05 and ensures that for a curve right the left shoulder will rise and the right shoulder will drop. The opposite is true for a curve left.

A special subroutine has been included for outputting the station in a form familiar to civil engineers and surveyors. The global label "STA" was used for the subroutine so that it could be used with other programs if desired. The subroutine takes the numerical value of the station (which must be stored in register seven) and outputs it as "STA XXX+XX.XX". For instance, 52254.06 stored in R07 would be output as "STA 522+54.06".

01768 C

PROGRAM DESCRIPTION I

CONTINUATION PAGE

Page 4 of 25

Much of the "Vert Align" section of the input phase of the SUPER program, and also the sections for calculation of the centerline elevation, consist of portions of the HP Surveying Module "VERT" program that were either used exactly as they appear in that program or modified somewhat to better suit the needs of the SUPER program. Following is a list of the SUPER program lines that were used from the "VERT" program, along with the corresponding line number from the "VERT" program:

<u>SUPER Line Number</u>	<u>Corresponding "VERT" Line Number</u>
12-74	02-64
76-116	66-106
117-124	108-115
125-142	187-204
144-190	116-162
308	237
312-316	238-242
322-324	243-245
325-326	247-248
328-331	250-253
342-358	254-270
447-449	283-285
592-593	286-287
595	289
600	288
601	290

Since the SUPER program was set up to be used with or without the Surveying Module, XEQ commands have been used in the program in all cases. When the Surveying Module is in use, those XEQ commands that call for one of the subroutine sets contained in the Module would execute faster if changed to XROM. This can be easily accomplished by deleting the particular line and then replacing it; using the XEQ key with the Module in place will give the XROM command in the new line.

E 1768 C

DEFINITION OF TERMS AND VARIABLES

P.C. - Point of curvature

P.I. - Point of intersection of tangents or grades

P.T. - Point of tangency

V - Design speed

D - Degree of horizontal curvature

R - Radius of horizontal curve, in feet

EL - Centerline elevation at station STA

STA - Station with centerline elevation EL

STA1 - Beginning station

GI - Beginning grade of vertical curve or grade of straight grade, in percent

EL1 - Beginning centerline elevation

G_n - Ending grade of vertical curve, in percent

L - Length of curve, in feet

EL0 - Centerline elevation of high or low point of vertical curve

ELI - Centerline elevation of vertical curve at P.I.

EL_i - Inside shoulder elevationEL_o - Outside shoulder elevationW_i - Width from centerline to inside shoulder, in feetW_o - Width from centerline to outside shoulder, in feet

e - Superelevation rate, ft/ft (interchangeable with S)

S - Superelevation rate, ft/ft (interchangeable with e)

c - Normal crown rate, ft/ft

f - Safe side friction factor

e+f - Centrifugal factor

RL - Superelevation runoff length, in feet

Z - Tangent runout (crown runoff) length, in feet

M - Multiplying factor for width of road

R_s - Distance from axis of superelevation rotation to farthest edge of pavement, in feet

r - Runoff ratio, or maximum relative grade between profiles of edge and centerline of a 24' pavement

W_L - Width from centerline to left shoulder, in feetW_R - Width from centerline to right shoulder, in feetW_T - Shoulder to shoulder width, in feet

B.F.S.- Begin full superelevation

E.F.S.- End full superelevation

FORMULAS FOR VERTICAL ALIGNMENT

For straight grade and beginning grade prior to vertical PC station:

$$(1) EL = (STA - STA_1)(G_1/100) + EL_1$$

For vertical curves, when L and PC are known:

$$(2) [(G_n - G_1)/200L] (STA - PC)^2 + (G_1/100)(STA - PC) + (EL_1 - EL) = 0$$

When L and PI are known:

$$(3) PC = PI - (L/2) \text{ Substitute in equation 2}$$

When EL₀ and PC are known:

$$(4) L = 200(EL_1 - EL_0)(G_n - G_1)(1/G_1^2)$$

When EL₀ and PI are known:

$$(5) L = 200(EL_1 - EL_0)(G_n - G_1)(1/G_n G_1)$$

When a specified point and PC are known:

$$(6) L = \{(STA - PC)^2 / [(G_1/100)(STA - PC) - (EL - EL_1)]\} [200 / (G_1 - G_n)]$$

When a specified point and PI are known:

$$(7) (L^2/4) + \{(STA - PI) - [200 / (G_1 - G_n)] [(G_1/100)(STA - PI) - (EL - EL_1)]\} L + (STA - PI)^2 = 0$$

$$(8) EL_1 = EL_1 - (G_1/100)(L/2)$$

For a station beyond the vertical PT station:

$$(9) EL = [L(G_n + G_1)/200] + [STA - (PC + L)] (G_n/100) + EL_1$$

FORMULAS FOR HORIZONTAL ALIGNMENT

$$(10) PT = PC + L$$

$$(11) R = 5729.58/D$$

$$(12) f = 0.09 + (100 - V)/1000$$

$$(13) e + f = \sqrt{15}R$$

$$(14) r = 1/[2.5(V - 30) + 150]$$

$$(15) M = [(R_s - 12)/24] + 1$$

$$(16) RL = (12eM)/r$$

$$(17) Z = (12cM)/r$$

FORMULAS FOR SHOULDER ELEVATIONS

For Case 1:

$$(18) EL_i = EL - W_i c$$

$$(19) EL_0 = EL - W_0 c$$

For Case 2:

$$(20) EL_i = EL - W_i c$$

$$(21) EL_0 = EL_i + cW_T [STA - (PC - 2RL/3 - Z)]/2Z$$

or

$$(21a) EL_0 = EL_i + cW_T [(PT + 2RL/3 + Z) - STA]/2Z$$

FORMULAS FOR SHOULDER ELEVATIONS (CONT'D)

For Case 3:

(22) $EL_i = EL - \{eW_i [STA - (PC - 2RL/3)] / RL\}$

(23) $EL_o = EL + \{eW_o [STA - (PC - 2RL/3)] / RL\}$

or


(22a) $EL_i = EL - [eW_i (PT + 2RL/3 - STA) / RL]$

(23a) $EL_o = EL + [eW_o (PT + 2RL/3 - STA) / RL]$

For Case 4:

(24) $EL_i = EL - eW_i$

(25) $EL_o = EL + eW_o$

SUPERELEVATION RUNOFF ON CENTER CROWN ROADWAY
(WHEN AXIS OF ROTATION IS CENTERLINE OF ROADWAY)

01768 C

PROGRAM DESCRIPTION I

CONTINUATION PAGE

Page 8 of 25

VALUES FOR DESIGN SUPERELEVATION RATE RELATED
TO DESIGN SPEED AND HORIZONTAL CURVATURE

D	V=30mph	V=40mph	V=50mph	V=60mph	V=65mph	V=70mph
0° 15'	NC	NC	NC	NC	NC	NC
0° 30'	NC	NC	NC	RC	RC	RC
0° 45'	NC	NC	RC	0.024	0.027	0.029
1° 00'	NC	RC	0.023	0.032	0.035	0.039
1° 30'	RC	0.021	0.033	0.046	0.052	0.058
2° 00'	RC	0.028	0.042	0.058	0.066	0.074
2° 30'	0.021	0.034	0.051	0.069	0.077	0.086
3° 00'	0.025	0.040	0.059	0.079	0.087	0.094
3° 30'	0.029	0.046	0.067	0.087	0.093	0.099
4° 00'	0.033	0.051	0.073	0.093	0.098	0.100
5° 00'	0.040	0.061	0.084	0.099	0.100	0.100
6° 00'	0.046	0.070	0.092	0.100	$D_{max} = 4.5^\circ$	$D_{max} = 4.0^\circ$
7° 00'	0.053	0.077	0.098	$D_{max} = 5.5^\circ$		
8° 00'	0.059	0.084	0.100			
9° 00'	0.064	0.089		0.100		
10° 00'	0.068	0.093		$D_{max} = 8.5^\circ$		
11° 00'	0.073	0.097				
12° 00'	0.077	0.099				
13° 00'	0.080	0.100				
14° 00'	0.083	0.100				
16° 00'	0.089		$D_{max} = 13.5^\circ$			
18° 00'	0.093					
20° 00'	0.097					
22° 00'	0.099					
24° 00'	0.100					
	0.100					
		$D_{max} = 25.0^\circ$				

$$e_{max} = 0.10$$

STANDARD SUPERELEVATION CRITERIA

D - Degree of curve

NC - Normal crown section

V - Assumed design speed

RC - Remove adverse crown, superelevate
at normal crown slope

e - Rate of superelevation

61768 C

CONTINUATION PAGE

VALUES FOR DESIGN SUPERELEVATION RATE RELATED
TO DESIGN SPEED AND HORIZONTAL CURVATURE

D	V=30mph	V=40mph	V=50mph	V=60mph	V=65mph	V=70mph
0° 15'	NC	NC	NC	NC	NC	NC
0° 30'	NC	NC	NC	RC	RC	RC
0° 45'	NC	NC	RC	0.022	0.025	0.029
1° 00'	NC	RC	0.021	0.029	0.033	0.038
1° 30'	RC	0.021	0.030	0.040	0.046	0.053
2° 00'	RC	0.027	0.038	0.051	0.057	0.065
2° 30'	0.021	0.033	0.046	0.060	0.066	0.073
3° 00'	0.025	0.038	0.053	0.067	0.073	0.078
3° 30'	0.028	0.043	0.058	0.073	0.077	0.080
4° 00'	0.032	0.047	0.063	0.077	0.079	0.080
5° 00'	0.038	0.055	0.071	0.080	0.080	D _{max} = 3.5°
6° 00'	0.043	0.061	0.077	0.080	D _{max} = 4.5°	
7° 00'	0.048	0.067	0.079	D _{max} = 5.0°		
8° 00'	0.052	0.071	0.080			
9° 00'	0.056	0.075	D _{max} = 7.5°			
10° 00'	0.059	0.077				
11° 00'	0.063	0.079				
12° 00'	0.066	0.080				
13° 00'	0.068	0.080				
14° 00'	0.070	D _{max} = 12.5°				
16° 00'	0.074					
18° 00'	0.077					
20° 00'	0.079					
22° 00'	0.080					
	0.080					
	D _{max} = 23.0°					

$$e_{\max} = 0.08$$

SUPERELEVATION CRITERIA WHERE
EXCEPTIONAL SNOW AND ICE CONDITIONS
ARE ENCOUNTERED IN CONJUNCTION WITH
MODERATE TO STEEP GRADES.

D - Degree of curve

NC - Normal crown section

V - Assumed design speed

RC - Remove adverse crown, superelevate

e - Rate of superelevation


at normal crown slope

01768 C PROGRAM DESCRIPTION II

Sample Problem (Sketch If Desired)

Calculate roadway elevations
for the given alignment
from Sta. 521+00 to
Sta. 536+00

Design Speed=40 mph



TYPICAL SECTION

TYPICAL SECTION

5.18%

NOTE: Substandard S.
Should be 0.07
based on crit-
eria.

SOLUTION:

Input	Function	Display	Comments
	XEO SIZE 025		Minimum size
	XEO SUPER	VERT ALIGN	Begin vertical alignment input
		CURVE?	
Y	R/S	PC=?	
52500	R/S	EL=?	
1361.43	R/S	GRADE BEG%=?	
1.65	R/S	GRADE END%=?	
5.518	R/S	L=?	
1000	R/S	HORIZ ALIGN	Begin horizontal alignment input
		CV R OR L?	
R	R/S	PC=?	
52254.06	R/S	L=?	
716.25	R/S	DESIGN SP=?	
40	R/S	LT SH W=?	
23	R/S	RT SH W=?	
26	R/S	SUPER=?	
.04	R/S	CROWN=?	
.02	R/S	RL=130/Z=65	Display runoff and runout lengths
	H	D=?	Check safe speed
8	R/S	SAFE SP=44.55	
52100	G	STA 521+0.00	
	R/S *	LT SH EL=1367.57	
	R/S *	C.L. EL=1368.03	
	R/S *	RT SH EL=1367.51	
100	H	LAST STA=?	Set spacing interval and last station for output
53600	R/S	STA 522+0.00	
	R/S *	LT SH EL=1366.60	
	R/S *	C.L. EL=1366.38	
	R/S *	RT SH EL=1365.86	
	R/S *	STA 523+0.00	
	R/S *	LT SH EL=1365.65	
	R/S *	C.L. EL=1364.73	
	R/S *	RT SH EL=1363.69	
	R/S *	STA 524+0.00	
	R/S *	LT SH EL=1364.00	

*Not required when using printer

01768 C

(CONTINUATION PAGE)

Input	Function	Display	Comments
	R/S *	C.L. EL=1363.08	
	R/S *	RT SH EL=1362.04	
	R/S *	STA 525+0.00	
	R/S *	LT SH EL=1362.35	
	R/S *	C.L. EL=1361.43	
	R/S *	RT SH EL=1360.39	
	R/S *	STA 526+0.00	
	R/S *	LT SH EL=1361.06	
	R/S *	C.L. EL=1360.14	
	R/S *	RT SH EL=1359.10	
	R/S *	STA 527+0.00	
	R/S *	LT SH EL=1360.48	
	R/S *	C.L. EL=1359.56	
	R/S *	RT SH EL=1358.52	
	R/S *	STA 528+0.00	
	R/S *	LT SH EL=1360.63	
	R/S *	C.L. EL=1359.71	
	R/S *	RT SH EL=1358.67	
	R/S *	STA 529+0.00	
	R/S *	LT SH EL=1361.48	
	R/S *	C.L. EL=1360.56	
	R/S *	RT SH EL=1359.52	
	R/S *	STA 530+0.00	
	R/S *	LT SH EL=1362.54	
	R/S *	C.L. EL=1362.14	
	R/S *	RT SH EL=1361.62	
	R/S *	STA 531+0.00	
	R/S *	LT SH EL=1364.08	
	R/S *	C.L. EL=1364.43	
	R/S *	RT SH EL=1363.91	
	R/S *	STA 532+0.00	
	R/S *	LT SH EL=1366.98	
	R/S *	C.L. EL=1367.44	
	R/S *	RT SH EL=1366.92	
	R/S *	STA 533+0.00	
	R/S *	LT SH EL=1370.71	
	R/S *	C.L. EL=1371.17	
	R/S *	RT SH EL=1370.65	
	R/S *	STA 534+0.00	
	R/S *	LT SH EL=1375.15	
	R/S *	C.L. EL=1375.61	
	R/S *	RT SH EL=1375.09	
	R/S *	STA 535+0.00	
	R/S *	LT SH EL=1380.31	
	R/S *	C.L. EL=1380.77	
	R/S *	RT SH EL=1380.25	
	R/S *	STA 536+0.00	
	R/S *	LT SH EL=1385.83	
	R/S *	C.L. EL=1386.29	
	R/S *	RT SH EL=1385.77	
52254.06	G	STA 522+54.06	Calculate elevations at horizontal
	R/S *	LT SH EL=1366.10	P.C. station
	R/S *	C.L. EL=1365.49	
	R/S *	RT SH EL=1364.79	


*Not required when using printer

01768 C PROGRAM DESCRIPTION II

Page 12 of 25

Sample Problem (Sketch If Desired)

Calculate roadway elevations
for the given alignment
from Sta. 105+00 to 115+00.
Design speed=50 mph



3° Cv. Lt.
Δ: 8°30'
T: 141.83'
L: 283.33'
R: 1909.86'
S: 0.059%

P.C. 105+00

P.T. 107+20

P.C. 111+00

P.T. 113+83.33

P.C. 115+00

110

115

2°30' Cv. Rt
Δ: 5°30'
T: 110.08'
L: 220.00'
R: 2291.83'
S: 0.051%

HORIZONTAL ALIGNMENT

948.50
105

970
960
950

+2.4%

VERTICAL ALIGNMENT

110

115

SOLUTION:

Input	Function	Display	Comments
	XEO SIZE 025		Minimum size
	XEO SUPER	VERT ALIGN	Begin vertical alignment input
N	R/S	CURVE?	
10500	R/S	STA1=?	
948.5	R/S	EL1=?	
2.4	R/S	GRADE%=?	
		HORIZ ALIGN	Begin horizontal alignment input
		CV R OR L?	
R	R/S	PC=?	
10500	R/S	L=?	
220	R/S	DESIGN SP=?	
50	R/S	LT SH W=?	
17	R/S	RT SH W=?	
17	R/S	SUPER=?	
.051	R/S	CROWN=?	
.02	R/S	RL=150/Z=59	Display runoff and runout lengths
	I	D=?	Check safe speed
2.5	R/S	SAFE SP=75.72	
10500	G	STA 105+0.00	
	R/S *	LT SH EL=949.08	
	R/S *	C.L. EL=948.50	
	R/S *	RT SH EL=947.92	
100	I	LAST STA=?	Set spacing interval and last station for output
10900	R/S	STA 106+0.00	
	R/S *	LT SH EL=951.77	
	R/S *	C.L. EL=950.90	
	R/S *	RT SH EL=950.03	
	R/S *	STA 107+0.00	
	R/S *	LT SH EL=953.99	
	R/S *	C.L. EL=953.30	
	R/S *	RT SH EL=952.61	
	R/S *	STA 108+0.00	
	R/S *	LT SH EL=955.82	
	R/S *	C.L. EL=955.70	
	R/S *	RT SH EL=955.36	

*Not required when using printer

01768 C

(CONTINUATION PAGE)

Input	Function	Display	Comments
	R/S * STA 109+0.00		
	R/S * LT SH EL=957.76		
	R/S * C.L. EL=958.10		
	R/S * RT SH EL=957.76		
10720	G STA 107+20.00		Calculate elevations at PT station
	R/S * LT SH EL=954.36		
	R/S * C.L. EL=953.78		
	R/S * RT SH EL=953.20		
	XFO 99 HORIZ ALIGN		Same vertical alignment: new
	CV R OR L?		horizontal curve
L	R/S PC=?		
11100	R/S L=?		
283.33	R/S DESIGN SP=?		
50	R/S LT SH W=?		
17	R/S RT SH W=?		
17	R/S SUPER=?		
.059	R/S CROWN=?		
.02	R/S RL=170/Z=58		Display runoff and runout lengths
11000	G STA 110+0.00		
	R/S * LT SH EL=960.16		
	R/S * C.L. EL=960.50		
	R/S * RT SH EL=960.58		
100	I LAST STA=?		Set spacing interval and last station for output
11500	R/S STA 111+0.00		
	R/S * LT SH EL=962.23		
	R/S * C.L. EL=962.90		
	R/S * RT SH EL=963.57		
	R/S * STA 112+0.00		
	R/S * LT SH EL=964.30		
	R/S * C.L. EL=965.30		
	R/S * RT SH EL=966.30		
	R/S * STA 113+0.00		
	R/S * LT SH EL=966.70		
	R/S * C.L. EL=967.70		
	R/S * RT SH EL=968.70		
	R/S * STA 114+0.00		
	R/S * LT SH EL=969.53		
	R/S * C.L. EL=970.10		
	R/S * RT SH EL=970.67		
	R/S * STA 115+0.00		
	R/S * LT SH EL=972.16		
	R/S * C.L. EL=972.50		
	R/S * RT SH EL=972.48		
11383.33	G STA 113+83.33		Calculate elevations at PT station
	R/S * LT SH EL=969.03		
	R/S * C.L. EL=969.70		
	R/S * RT SH EL=970.37		

*Not required when using printer

01768C

USER INSTRUCTIONS

Page 14 of 25

SIZE: 025
(HP-41C)

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Enter the program			
2	For complete solution with elevations, begin the SUPER program. Or,		<input checked="" type="checkbox"/> SUPER	VERT ALIGN CURVE?
2a	To use as only as a check of design data, begin HA subroutine. Then go to step 7.		<input checked="" type="checkbox"/> HA	HORIZ ALIGN CV R OR L?
3	If calculating a vertical curve, input Y. Then go to step 4.	Y	<input checked="" type="checkbox"/> R/S	PC=?
3a	If calculating a grade, input N. Then call prompting and input beginning station, beginning elevation and grade. Then go to step 7.	N STA1 EL1 GRADE%	<input checked="" type="checkbox"/> R/S <input checked="" type="checkbox"/> R/S <input checked="" type="checkbox"/> R/S	STA1=? EL1=? GRADE%=? HORIZ ALIGN *** CV R OR L?
4	Input vertical PC station. If PC is not known, press <input checked="" type="checkbox"/> R/S and then input the vertical PI station.	PC — PI	<input checked="" type="checkbox"/> R/S <input checked="" type="checkbox"/> R/S <input checked="" type="checkbox"/> R/S	EL=? PI=? EL=?
5	Input elevation at PC or PI (whichever was input above) and beginning and ending grades.	EL GRADE BEG% GRADE END%	<input checked="" type="checkbox"/> R/S <input checked="" type="checkbox"/> R/S	GRADE BEG%=? GRADE END%=? L=?
6	Input length of vertical curve, if known. Or, if unknown, press <input checked="" type="checkbox"/> R/S and input elevation of high or low point if known. Or, if unknown, press <input checked="" type="checkbox"/> R/S and input a station and elevation through which the vertical curve passes.	L — EL0 — STA EL	<input checked="" type="checkbox"/> R/S <input checked="" type="checkbox"/> R/S	HORIZ ALIGN *** CV R OR L? EL0=? HORIZ ALIGN *** CV R OR L? STA=? EL=? HORIZ ALIGN *** CV R OR L?
7	If a curve right, input R. If a curve left, input L.	L or R	<input checked="" type="checkbox"/> R/S	PC=?
8	Input horizontal curve PC station.	PC	<input checked="" type="checkbox"/> R/S	L=?
9	Input length of horizontal curve.	L	<input checked="" type="checkbox"/> R/S	DESIGN SP=?
10	Input design speed.	DESIGN SP	<input checked="" type="checkbox"/> R/S	LT SH W=?
11	Input width to left shoulder.	LT SH W	<input checked="" type="checkbox"/> R/S	RT SH W=?
12	Input width to right shoulder.	RT SH W	<input checked="" type="checkbox"/> R/S	SUPER=?
13	Input superelevation rate.	SUPER	<input checked="" type="checkbox"/> R/S	CROWN=?

01768C USER INSTRUCTIONS

SIZE: 025
(HP-41C)

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
14	Input crown rate.	CROWN	R/S	RL= /Z=
15	To check the design speed against the safe speed for the input parameters, go to step 16. To calculate elevations at a specific station, go to step 17.			
16	Begin safe speed check and input degree of curve of horizontal curve. (Minutes must be input as a decimal of a degree. Ex. 2°45' would be 2.75)	D	H	D=?
17	Input station and calculate its elevations. Repeat step 17 for next station or, go to step 18 for automatic stationing.	STA	R/S	SAFE SP=
			R/S *	LT SH EL=
		:	R/S *	C.L. EL=
			R/S *	RT SH EL=
18	Input stationing interval and input last station desired for output; automatically calculate successive stations and elevations beginning from the current station.	INT	I	LAST STA=?
		LAST STA	R/S	STA +
			R/S *	LT SH EL=
			R/S *	C.L. EL=
			R/S *	RT SH EL=
			etc.	
19	For a totally new alignment, or for the same horizontal alignment with a change in vertical alignment, go to step 2.			
19a	For the same vertical alignment with a change in the horizontal alignment, XEQ 99 then go to step 7.		XEQ 99	HORIZ ALIGN CV R OR L?
	*** After completion of vertical alignment input, if using the same horizontal alignment from the previous solution, take the calculator out of ALPHA mode and go to step 15.		ALPHA	
	* This R/S not required when using printer.			

□ 67 □ 97 □ 41C

STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS	STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS
* 01	LBL "SUPER"			51	CLA		
	CF 21				ARCL 01		
	FS? 55				"� END%=?"		
	SF 21				PROMPT		
	SF 12				CF 22		
	"VERT ALIGN"		Begin input of vertical alignment parameters.		LBL 03		
	AVIEW				R4		
	PSE				STO 07		
	PSE				STO 00		Store beginning station (or vertical curve P.C. station)
10	CLD			60	CLX		
	CF 12				1 E2		
	XEQ "*IN"				/		
	ASTO 13				STO 03		
	ASTO 12				X<>Y		
	SF 02				LAST X		
	"GRADE"				/		
	ASTO 01				STO 02		
	"CURVE"				R4		
	XEQ "*YN"				STO 01		
20	FS? 10			70	FC? 01		
	GTO 08		If vertical curve, GTO 08, else input		GTO 10		
	SF 10		beginning station		CF 10		
	1		and elevation and		CLA		
	STO 13		then GTO 03.		ASTO 13		
	XEQ "*S"				GTO 99		
	XEQ "*EL"				LBL 10		
	CLA				"L=?"		
	ARCL 01				PROMPT		
	"�%=?"				FS?C 22		
30	PROMPT			80	GTO C		
	SF 01				0		
	0				STO 13		
	GTO 03				XEQ "*EL"		
*	LBL 08		For vertical curve input P.C. station		FS?C 22		
	CF 22		if known, else input		GTO a		
	"PC=?"		vertical P.I. station.		CLA		
	PROMPT				ASTO 13		
	FS?C 22				XEQ "*S"		
	GTO 09				FC?C 22		
40	"PI=?"			90	GTO 10		
	PROMPT				XEQ "*EL"		
	FC?C 22				LBL b		
	GTO 08				RCL 01		
	CF 02				-		
*	LBL 09				X<>Y		
	XEQ "*EL"		Input either vertical curve P.C.		RCL 00		
	CLA		elevation or P.I.		-		
	ARCL 01		elevation depend-		STO 08		
	"� BEG%=?"		ing on which is		RCL 08		
50	PROMPT		known.	100	RCL 02		

□ 67 □ 97 ☐ 41C

STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS	STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS
101	*			51	2		
	R↑				*		
	-				RCL 03		
	RCL 02				RCL 02		
	RCL 03				-		
	-				*		
	/				RCL 02		
	2				X↑2		
	*				/		
10	FS? 02			60	FS? 02		
	GTO 06				GTO 01		
	-				RCL 02		
	RCL 08				*		
	X↑2				RCL 03		
	4				/		
	1/X				*	LBL C	
	XEQ 04				STO 05		
	X < 0?				FS? 02		
	X<>Y				GTO 01		
20	GTO C			70	2		
*	LBL 06				/		
	/				RCL 00		
	*				X<>Y		
	GTO 01				-		
*	LBL 04				STO 00		
	/				STO 07		
	X <>Y				LAST X		
	LAST X				RCL 02		
	/				*		
30	2			80	ST-01		
	/				RCL 05		
	CHS			*	LBL 01		
	ENTER↑				RCL 03		
	ENTER↑				RCL 02		
	X↑2				-		
	R↑				X<>Y		
	-				STO 05		
	SQRT				/		
	-				STO 06		
40	LAST X			90	ADV		
	R↑			*	LBL 99		
	+				CLA		
	RTN				CF 21		
*	LBL a		Calculate length of vertical curve when the elevation of the low point or high point on the curve is known.		FS? 55		
	RCL 12				SF 21		
	STO 13				SF 12		
	RDN				"HORIZ ALIGN"		
	RCL 01				AVIEW		
	X<>Y				PSE		
50	-			200	PSE		

Store length of vertical curve.

Store algebraic difference of grades divided by length of vertical curve.

Begin input of horizontal alignment parameters.

□ 67 □ 97 □ 41C

STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS	STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS
201	CLD			51	12		Calculate width
	CF 12				-		multipling factor
	"CV R OR L?"			24			then use it and 1/r
	AON			/			to calculate runoff
	PROMPT			1			length and store.
	AOFF			+			
	ASTO X			*			
	"R"			STO 19			
	ASTO Y			RCL 14			
10	CLA			60	50		
	CF 05				-		Compare runoff
	X=Y?				X > Y?		length to minimum
	SF 05				GTO 13		standard. If RL is
	"PC=?"				X <= Y		larger, calculate
	PROMPT				25		rounded value of
	STO 15				MOD		RL that is closest
	"L=?"				2.5		to either a multi-
	PROMPT				X > Y?		ple of 25 or 10
	RCL 15				GTO 12		and store.
20	+			70	RCL 19		
	STO 16				10		
*	LBL "HA"				/		
	"DESIGN SP=?"				.5		
	PROMPT				+		
	STO 09				INT		
	30				10		
	-				*		
	2.5				GTO 13		
	*				*		
30	1.5 E2		Calculate 1/r and	80	LBL 12		
	+		store.		X <= Y		
	STO 14				RCL 19		
	"LT SH W=?"				-		
	PROMPT				CHS		
	STO 17				*		
	"RT SH W=?"				LBL 13		
	PROMPT				STO 19		
	STO 18				"CROWN=?"		
	"SUPER=?"				PROMPT		
40	PROMPT				STO 22		
	STO 10				*		
	RCL 14			90	RCL 10		
	*				/		
	12				.5		
	*				+		
	RCL 17		Compare shoulder		INT		
	RCL 18		widths and use		STO 20		
	X > Y?		larger value to		ADV		
	X <= Y		calculate width		ADV		
50	RDN		multipling factor		FIX 0		
					CF 29		
				300	"RL="		Display runoff

01768 C

PROGRAM LISTING

Page 19 of 25

 67 97 41C

STEP/ LINE	KEY ENTRY	KEY CODE (67/87 only)	COMMENTS	STEP/ LINE	KEY ENTRY	KEY CODE (67/87 only)	COMMENTS
301	ARCL 19			51	*		
	"F/"		length and Z distance.		2		
	"FZ="				/		
	ARCL X				RCL 02		
	AVIEW				+		
	ADV				*		
	RTN				RCL 01		
*	LBL G				+		
	FS? 10			*	LBL 14		
10	CF 01			60	STO 21		Store centerline elevation.
	FIX 2				ADV		
	FS?C 22				SF 21		
	GTO 02		If a station has been entered, then GTO 02 to calculate centerline elevation, else add interval to previous-		RCL 07		
	RCL 04		station for new		RCL 15		
	RCL 07		station and check		RCL 19		
	+		to see if station is within limits of vertical curve.		2		
	RCL 00				*		
	X>Y?				3		
	SF 01				/		
20	FS? 01			70	-		
	X<>Y				RCL 20		
	FS? 01				-		
	GTO 05				X<=Y?		
	RCL 05	-			GTO 07		
	+			*	LBL 27		
	X<=Y?				XEQ 98		
	GTO 21				XEQ 17		
	X<>Y				XEQ 15		
	GTO 05				XEQ 22		
*	30	LBL 02		80	XEQ 18		
	SF 00				GTO 19		
	RCL 00			*	LBL 21		
	X>Y?				X<>Y		
	SF 01			*	LBL 23		
	RCL 05				STO 07		
	+				X<>Y		
	X<>Y				-		
	X<=Y?				RCL 03		
	GTO 05				*		
40	FS? 10			90	RCL 03		
	GTO 23				RCL 02		
*	LBL 05				+		
	STO 07				2		
	RCL 00				/		
	-				RCL 05		
	ENTER↑				*		
	ENTER↑				+		
	RCL 06				RCL 01		
	FS? 01				+		
50	CLX			400	GTO 14		

01768 C

PROGRAM LISTING

Page 20 of 24

□ 67 □ 97 ☐ 41C

STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS	STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS
* 401	LBL 07			51	RCL 21		Calculate left shoulder elevation for normal crown section.
	RCL 20				RCL 17		
	2				RCL 22		
	*				*		
	+				-		
	X<=Y?				RTN		
	GTO 16			*	LBL 18		Calculate right shoulder elevation for normal crown section.
	RCL 20				RCL 21		
	2				RCL 18		
10	*			60	RCL 22		
	-				*		
	-				-		
*	LBL 28				RTN		
	RCL 20			*	LBL 15		
	2				CLA		
	*				"LT"		
	/			*	LBL 20		Display left shoulder elevation.
	RCL 17				"↑ SH EL"		
	RCL 18				ARCL X		
20	+			70	AVIEW		
	RCL 22				RTN		
	*			*	LBL 22		Display centerline elevation.
	*				"C.L. EL"		
	XEQ 98	-			ARCL 21		
	X<>Y				AVIEW		
	FS? 05				RTN		
	GTO 11			*	LBL 16		
	XEQ 17				X<>Y		
	XEQ 15				RCL 15		
30	XEQ 22			80	RCL 19		
	+				3		
	GTO 19				/		
*	LBL 11				+		
	XEQ 18				X<=Y?		If station is beyond start of full superelevation, (PC+RL/3), GTO 00, else calculate amount of super of left shoulder.
	+				GTO 00		
	XEQ 15				-		
	XEQ 22				RCL 19		
	XEQ 18				/		
*	LBL 19				1		
40	CLA			90	+		
	"RT"			*	LBL 25		
	XEQ 20				RCL 17		
	RCL 07			*	RCL 10		
	RCL 23			*	FC? 05		If curve left, subtract from centerline elevation, else add to centerline elevation.
	X<=Y?				CHS		
	SF 00				RCL 21		
	FS? 00				+		
	STOP			500	XEQ 98		
	GTO G						
*	50	LBL 17					

01768 C

□ 67 □ 97 ☐ 41C

PROGRAM LISTING

Page 21 of 2

STEP/ LINE	KEY ENTRY	KEY CODE (67/87 only)	COMMENTS	STEP/ LINE	KEY ENTRY	KEY CODE (67/87 only)	COMMENTS
501	X<->Y			51	RCL 19		
	XEQ 15				/		
	XEQ 22				CHS		
	RCL 21				GTO 25		
	-			*	LBL 26		
	RCL 17				RCL 20		
	/				2		
	RCL 18				*		
	*				+		
10	CHS		Calculate superelevation of right shoulder and add to centerline elevation with opposite sign of left shoulder superelevation.	60	X<=Y?		If station is beyond end of tangent runout(PT+2RL/3+Z), GTO 27, else calculate ΔE .
	RCL 21				GTO 27		
	+				-		
	GTO 19				CHS		
*	LBL 00				GTO 28		
	X<->Y			*	LBL "STA"		SUBROUTINE TO OUTPUT STATION IN
	RCL 16			*	LBL 98		COMMON ENGINEERING
	RCL 19				CLA		AND SURVEYING FORMAT OF "STA XXX+XX XX"
	3				RCL 07		
	/				FIX 0		
20	-			70	1 E2		
	X<=Y?		If station is beyond end of full superelevation (PT-RL/3), GTO 24		/		
	GTO 24				ENTER↑		
	XEQ 98				ENTER↑		
	RCL 21				CF 29		
	RCL 17				INT		
	RCL 10				ARCL X		
	*				ASTO 11		
	FC? .05				CLA		
	CHS				X<->Y		
30	+			80	RCL T		
	XEQ 15				X<->Y		
	XEQ 22				FIX 2		
	RCL 21				FRC		
	RCL 18				1 E2		
	RCL 10				*		
	*				"STA"		
	FS? .05				ARCL 11		
	CHS				"F+"		
	+				ARCL X		
40	GTO 19			90	AVIEW		
*	LBL 24				RTN		
	RCL 19			*	LBL I		Store interval for
	+				CF 00		automatic spacing
	RCL 20				ADV		of points.
	-				STO 04		
	X<=Y?				ADV		
	GTO 26				"LAST STA=?"		Store last station
	RCL 20				PROMPT		required for output.
	+				STO 23		
50	-			600	CF 22		

01768C

PROGRAM LISTING

Page 22 of 25

 67 97 41C

STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS	STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS
601	GTO G			51			
*	LBL H		SUBROUTINE TO CHECK				
	ADV		DESIGN SPEED				
	ADV		AGAINST SAFE SPEED				
	"D=?"		BASED ON DESIGN				
	PROMPT		PARAMETERS OF SUPER				
	STO 24		ELEVATION, DEGREE				
*	LBL 29		OF CURVE, AND SAFE				
	1.9 E2		SIDE FRICTION FAC-				
10	RCL 09		TOR.	60			
	-						
	RCL 10						
	1 E3						
	*						
	+						
	1.164 E-2						
	/						
	RCL 24						
	/						
20	SQRT			70			
	RCL 09						
	-						
	ENTER↑						
	ENTER↑						
	ABS						
	.5						
	X>Y?						
	GTO 30						
	X<>Z						
30	ST+09			80			
	GTO 29						
*	LBL 30						
	ADV						
	FIX 2						
	"SAFE SP="						
	ARCL 09						
	AVIEW						
	END						
40				90			
50				00			

01768C

PROGRAM LISTING

Page 23 of 25

 67 97 41C

SUBROUTINE SET

STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS	STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS
*	01	LBL "★IN"	Initialize program	51			
		SF 27					
		SF 21					
		SF 24					
		SF 29					
		FIX 2					
		CLA					
		CF 00					
		CF 01					
10		CF 02		60			
		CF 22					
		RTN					
*		LBL "★YN"					
		"†?"					
		AON					
		PROMPT					
		AOFF					
		ASTO X					
		"Y"					
20		ASTO Y		70			
		CLA					
		CF 10					
		X=Y?	If answer is yes, set flag 10.				
		SF 10					
		RTN					
*		LBL "★EL"					
		"EL"					
		GTO 01					
*		LBL "★S"					
30		"STA"		80			
*		LBL 01					
		CF 29					
		FIX 0					
		ARCL 13					
		"†="					
		FIX 2					
		SF 29					
		FS? 10					
		"†?"					
40		FC? 10		90			
		ARCL X					
		FC? 10					
		RTN					
		PROMPT					
		RTN					
50				00			

REGISTERS, STATUS, FLAGS, ASSIGNMENTS


DATA REGISTERS		STATUS			
		SIZE ENG DEG	025 FIX 0,2 RAD	TOT. REG. SCI GRAD	206* ON X OFF
00	P.C. station of vertical curve or STA1 of straight grade	*190 if HP Surveying Module used			
01	P.C. elevation of vertical curve or STA1 elevation of straight grade				
02	Decimal form of straight grade or beginning grade for vertical curve				
03	Decimal form of ending grade of vertical curve	#	INIT S/C	SET INDICATES	CLEAR INDICATES
04	Interval between points	00	C	Interval execution	No interval execution
05	Length of vertical curve	01	C	Station outside limits of the vertical curve	Station within limits of the vertical curve
06	(R03-R02)/R05	02	C	Vertical P.C. known	Vertical P.C. unknown
07	Current station	05	C	Horizontal curve right	Horizontal curve left
08	Distance of station from vertical P.C. (Used when L is unknown, and the station and elevation of a point on the vertical curve are input.)	10	C	Vertical curve	No vertical curve
09	Design speed	12	C	Double wide print	Normal width print
10	Superelevation rate (ft/ft)	21	S	Printer enabled	Printer disabled
11	Alpha designation of hundreds portion of station in R07	22	C	Digit entered	No digit entered
12	Blank (Used to reset R13)	24	S	Ignore range error	Display range error
13	Blank, 0, or 1	27	S	USER mode in	USER mode out
14	1/r	29	S	Digit grouping	Suppresses decimal point in FIX 0 mode
15	P.C. station of horizontal curve	55	C	Printer connected	No printer
16	P.T. station of horizontal curve				
17	Left shoulder width				
18	Right shoulder width				
19	Superelevation runoff length (RL)				
20	Tangent runout length (Z)				
21	Centerline elevation of station in R07				
22	Crown rate (ft/ft)				
23	Last station to calculate				
24	Degree of curve of horizontal curve				
ASSIGNMENTS					
FUNCTION	KEY	FUNCTION	KEY	FUNCTION	KEY

01768C


KEYBOARD CARD LABELING

Page 25 of 25

KEYBOARD




SYSTEM
CONFIGURATION



*or Survey Mod

CARD



Scan Copyright ©
The Museum of HP Calculators
www.hpmuseum.org

Original content used with permission.

Thank you for supporting the Museum of HP
Calculators by purchasing this Scan!

Please do not make copies of this scan or
make it available on file sharing services.